

Bedingte Formatierung

02.07.2019

Beispiel 1: Einfache Suchbox

	A	B	C	D	E	F
1	Erfolgreiche Filme					
2	Quelle: https://de.wikipedia.org/wiki/Liste_erfolgreicher_Filme					
3						
4						
5	Suche: <input type="text" value="Walt"/>					
6						
7	Pl.	Deutscher Titel	Originaltitel	Jahr	Einspielergebnis	Filmverleih (US)
8	1	Avatar – Aufbruch nach Pandora	Avatar	2009	2788,0	20th Century Fox
9	2	Titanic	Titanic	1997	2186,8	Paramount Pictures
10	3	Jurassic World	Jurassic World	2015	1665,2	Universal Pictures
11	4	Marvel's The Avengers	The Avengers	2012	1519,6	Walt Disney Studios
12	5	Fast & Furious 7	Furious 7	2015	1511,7	Universal Pictures
13	6	Avengers: Age of Ultron	Avengers: Age of Ultron	2015	1402,8	Walt Disney Studios
14	7	Harry Potter und die Heiligtümer des Todes – Teil 2	Harry Potter and the Deathly Hallows: Part 2	2011	1341,5	Warner Bros.
15	8	Die Eiskönigin – Völlig unverfroren	Frozen	2013	1274,2	Walt Disney Studios
16	9	Iron Man 3	Iron Man 3	2013	1215,4	Walt Disney Studios
17	10	Minions	Minions	2015	1152,7	Universal Pictures
18	11	Transformers 3 – Die dunkle Seite des Mondes	Transformers: Dark of the Moon	2011	1123,8	Paramount Pictures
19	12	Der Herr der Ringe: Die Rückkehr des Königs	The Lord of the Rings: The Return of the King	2003	1119,9	New Line Cinema
20	13	James Bond 007 – Skyfall	Skyfall	2012	1108,6	Metro-Goldwyn-Mayer

Regel	Format	Angewendet auf
=UND(NICHT(ISTLEER(\$C\$5));SUCHEN(\$C\$5;\$A8&\$B8&\$C8&\$D8&\$E8&\$F8))	AaBbCcYyZz	=\$A\$8:\$F\$107

Beispiel 2: Felder in Formularen dynamisch ein- und ausblenden

	A	B	C	D	E	F	G	H	I	J
1										
2										
3	Vertragliche Zeiten									
4										
5	Arbeitswoche: <input type="text" value="5-Tage-Woche"/>									
6										
7	Wöchentliche tarifliche Arbeitszeit in Stunden: 40,10									
8										
9	Persönliche wöchentliche Arbeitszeit in Stunden: <input type="text" value="38,08"/>									
10										
11	Persönl. Arbeitstage in der Ferienzeit: <input type="text" value="15,0"/>									
12										
13										
14										

Regel	Format	Angewendet auf
=\$F\$5<>"5-Tage-Woche"	AaBbCcYyZz	=\$H\$3:\$J\$14;\$G\$5
=\$F\$5<>"5-Tage-Woche"	AaBbCcYyZz	=\$I\$5;\$I\$7;\$I\$9;\$I\$11;\$I\$13
=\$F\$5<>"5-Tage-Woche"	AaBbCcYyZz	=\$H\$4:\$J\$14
=\$N\$6=FALSCH	AaBbCcYyZz	=\$F\$5

Beispiel 3: Entfernungsrechner

	A	B	C	D	E	F	G	H	I	J
1										
2		Von:	Dresden					Entfernungskilometer:		
3		Nach:	Erfurt					230		
4										
5										
6										
7	Aachen		562	538	647	371	745	642	453	124
8	Augsburg		562	244	588	713	580	479	412	599
9	Bayreuth		538	244	344	577	331	231	189	485
10	Berlin		647	588	344	365	123	224	297	483
11	Bremen		371	713	577	365	489	482	354	250
12	Cottbus		745	580	331	123	489	134	314	617
13	Dresden		642	479	231	224	482	134	230	585
14	Erfurt		453	412	189	297	354	230		377
15	Essen		124	599	485	483	250	585	377	
16	Frankfurt a.M.		247	362	258	563	444	590	259	250

Regel	Format	Angewendet auf
=UND(\$A7=\$B\$2;\$B\$6=\$B\$3)	AaBbCcYyZz	=\$B\$7:\$AO\$46
=B6=\$B\$3	AaBbCcYyZz	=\$B\$6:\$AO\$6
=A7=\$B\$2	AaBbCcYyZz	=\$A\$7:\$A\$46
=UND(\$B\$6=\$B\$3;ZEILEN(\$B\$7:B7)<=VERGLEICH(\$B\$2;\$A\$7:\$A\$46;0))	AaBbCcYyZz	=\$B\$7:\$AO\$46
=UND(\$A7=\$B\$2;SPALTEN(\$B\$7:B7)<=VERGLEICH(\$B\$3;\$B\$6:\$AO\$6;0))	AaBbCcYyZz	=\$B\$7:\$AO\$46

Beispiel 4: Ungültige Daten bei abhängigen Dropdownlisten kennzeichnen

	A	B	C	D	E	F	G	H	I	J	K	L
1	Hersteller	Modell					Hersteller		Audi	BMW	Mercedes	Volkswagen
2	Mercedes	A3					Audi		A1	1er-Reihe	A-Klasse	Polo
3							BMW		A2	2er-Reihe	B-Klasse	Beetle
4							Mercedes		A3	3er-Reihe	C-Klasse	Golf
5							Volkswagen		A4	5er-Reihe	E-Klasse	Passat
6									A6	6er-Reihe	S-Klasse	Touran
7										7er-Reihe		Tiguan
8										M-Serie		Sharan
9										Z-Serie		Phaeton
10										X-Serie		
11												

Regel	Format	Angewendet auf
=ODER(UND(A2="";B2<>"");UND(B2<>"";ISTFEHLER(SVERWEIS(B2;INDIREKT(A2);1; FALSCH))))	AaBbCcYyZz	=\$B\$2:\$B\$11

Beispiel 5: Regeln per Schalter aktivieren/deaktivieren

	A	B	C	D	E	F	G
1							
2		Marketingplan					
3							
6							
7							
8							
11		AUFGABE	STATUS	BESITZER	ZUGEWIESEN AN	STARTDATUM	ENDDATUM
12		Produktanalyse	In Bearbeitung	Johann C.	Johann C.	01.07.2018	01.08.2018
13		Storyboards entwerfen	Verzögert	Mark M.	Johann C.	15.07.2018	15.08.2018
14		Entwurf des Storyboards überprüfen	Erledigt	Mark M.	Johann C.	01.08.2018	20.08.2018
15		Analyse der Recherchen, Phase I	Verzögert	Karl A.	Andreas L.	01.06.2018	01.07.2018
16		Erstellung der Anzeigeninhalte, Phase I	Verzögert	Karl A.	Timon S.	01.09.2018	15.09.2018
17		Definitionen der Produktanforderungen	Verzögert	Karl A.	Viviane A.		
18		Spezifikationen für die Entwicklung des Prototyps	In Bearbeitung	Karl A.	Gabriele F.	12.09.2018	25.09.2018
19		Qualitätskontrolle, Fortschrittsberichte	In Bearbeitung	Mark M.	Karl A.	01.07.2018	01.10.2018
20		Storyboards erstellen	Nicht begonnen	Mark M.	Johann C.	15.07.2018	15.08.2018
21		Storyboards mit Grafikern durchgehen	Verzögert	Mark M.	Johann C.	01.08.2018	20.08.2018
22		Analyse der Recherchen, Phase II	Erledigt	Karl A.	Andreas L.	01.06.2018	01.07.2018
23		Erstellung der Anzeigeninhalte, Phase II	Nicht begonnen	Karl A.	Timon S.	01.09.2018	15.09.2018

Regel	Format	Angewendet auf
=UND(\$C12="Nicht begonnen";\$D\$7="EIN")	AaBbCcYyZz	=\$B\$12:\$G\$23
=UND(\$C12="In Bearbeitung";\$E\$7="EIN")	AaBbCcYyZz	=\$B\$12:\$G\$23
=UND(\$C12="Verzögert";\$F\$7="EIN")	AaBbCcYyZz	=\$B\$12:\$G\$23
=UND(\$C12="Erledigt";\$G\$7="EIN")	AaBbCcYyZz	=\$B\$12:\$G\$23

Regelmanager

- Jede Formatierungsregel liefert einen Wahrheitswert:
WAHR -> Regel wird angewendet
FALSCH -> Regel wird nicht angewendet.
- Eine Regel, die sich an höherer Stelle in der Liste befindet, besitzt Vorrang vor einer Regel weiter unten in der Liste.
- Standardmäßig werden neue Regeln stets am Anfang der Liste hinzugefügt und besitzen daher Vorrang, sodass auf deren Reihenfolge geachtet werden muss.

Falls es für eine Zelle oder einen Zellbereich mehrere Regeln gibt, wird unterschieden:

- Regeln lösen keinen Konflikt aus

Wenn beispielsweise eine Zelle nach einer Regel fett formatiert wird und dieselbe Zelle nach einer anderen Regel mit Rot formatiert wird, wird die Zelle sowohl fett als auch rot formatiert. Da kein Konflikt zwischen den beiden Formaten auftritt, werden beide Regeln angewendet.

- Regeln lösen Konflikt aus

Beispielsweise wird durch eine Regel eine Zellschriftfarbe auf Rot festgelegt, und durch eine andere Regel wird eine Zellschriftfarbe auf Grün festgelegt. Da die beiden Regeln einen Konflikt auslösen, kann nur eine angewendet werden. Die angewendete Regel entspricht der mit der höheren Stellung in der Rangfolge (weiter oben in der Liste im Dialogfeld).

- Über den „Anhalten“-Haken kann die Regelauswertung an einer bestimmten Stelle gestoppt werden. Damit werden aber nur solche nachfolgenden Regeln ignoriert, die für die gleiche Zelle oder den gleichen Zellbereich zu einem WAHR führen würden

Quelle:

<https://support.office.com/de-de/article/Verwalten-der-Rangfolge-von-Regeln-zur-bedingten-Formatierung-E09711A3-48DF-4BCB-B82C-9D8B8B22463D>

Stolpersteine

- Beim Anlegen der Regel auf die aktive Zelle achten: Zellbezüge in Formeln sind immer relativ zur aktiven Zelle zu sehen
- Reihenfolge der Regeln ist wichtig:
 - Bei Konflikten wird nur die ranghöhere (weiter oben stehende) Regel ausgeführt
 - Ohne Konflikte werden alle zutreffenden Regeln ausgeführt (Ausnahme: "Anhalten")
- Beim Übertragen von Formaten (Kopieren & Inhalte einfügen | Format) werden vorhandene bedingte Formatierungen überschrieben
- Zwischen unterschiedlichen Arbeitsmappen kopierte Formatierungsregeln können zu unerwünschten Datenverknüpfungen führen
- Bedingte Formatierungen machen jede betroffene Zelle volatil. Auswirkungen auf Performance!

Exkurs: Volatile Funktionen

BEREICH.VERSCHIEBEN

HEUTE

INDIREKT

INFO

JETZT

ZELLE (mit Parameter „Dateiname“)

ZUFALLSBEREICH

ZUFALLSZAHL